



United States
Department of
Agriculture

Agricultural
Research Service

Southern Regional
Research Center

Technical Report
June 2019
Revised February 2020

A Simple but Accurate Excel User-Defined Function to Calculate Studentized Range Quantiles for Tukey Tests

K. Thomas Klasson

The Agricultural Research Service (ARS) is the U.S. Department of Agriculture's chief scientific in-house research agency. Our job is finding solutions to agricultural problems that affect Americans every day from field to table. ARS conducts research to develop and transfer solutions to agricultural problems of high national priority and provide information access and dissemination of its research results.

The U.S. Department of Agriculture (USDA) prohibits discrimination in all its programs and activities on the basis of race, color, national origin, age, disability, and where applicable, sex, marital status, familial status, parental status, religion, sexual orientation, genetic information, political beliefs, reprisal, or because all or part of an individual's income is derived from any public assistance program. (Not all prohibited bases apply to all programs.) Persons with disabilities who require alternative means for communication of program information (Braille, large print, audiotape, etc.) should contact USDA's TARGET Center at (202) 720-2600 (voice and TDD). To file a complaint of discrimination, write to USDA, Director, Office of Civil Rights, 1400 Independence Avenue, S.W., Washington, D.C. 20250-9410, or call (800) 795-3272 (voice) or (202) 720-6382 (TDD). USDA is an equal opportunity provider and employer.

K. Thomas Klasson is a Supervisory Chemical Engineer at USDA-ARS, Southern Regional Research Center, 1100 Robert E. Lee Boulevard, New Orleans, LA 70124; email: thomas.klasson@ars.usda.gov

A Simple but Accurate Excel User-Defined Function to Calculate Studentized Range Quantiles for Tukey Tests

K. Thomas Klasson

U.S. Department of Agriculture
Agricultural Research Service
Southern Regional Research Center
New Orleans, Louisiana, USA

Technical Report
June 2019, Revised February 2020

Abstract

Microsoft Excel lacks a built-in function for calculation of Studentized range quantiles. Having these quantiles (Q) available in Excel is an advantage as it is easy to perform comparisons of means in post hoc tests as part of analysis of variance. Thus, an accurate, fast, and easily implemented user-defined function was developed for calculation of these quantiles. The developed procedure was about 200 times faster than other accurate Excel methods available and provided very low error rates over a wide range of probabilities.

Introduction

Experimentalists often engage in comparisons of means obtained under various treatments. Multiple comparison procedures (MCP's) are often carried out by statistical software packages that are less suitable for data storage and manipulation and may also require the user to learn a new computer programming skill (Kramer et al. (2016)). Microsoft Excel offers excellent data storage and manipulation but is very limited when it comes to statistical data analysis. The availability of Excel functions such as FDIST, FINV, TDIST, and TINV for the F - and t -distributions allows for some capability to conduct simple MCP's such as Fisher's Least Significant Difference (LSD) and the Scheffe tests. However, the lack of support for the Studentized range (Q) distribution does not allow for the Tukey Honest Significant Difference (HSD), Student-Newman-Keuls (SNK), or Ryan-Einot-Gabriel-Welsch Q (REGWQ) tests to be carried out. A comprehensive review of different MCP methods may be found in Day and Quinn (1989) who described, provided equations for, and evaluated many of the commonly used methods. While this reference is somewhat dated, it is an excellent review and is still cited in recent works (Franklin et al. (2018); Schroeder-Spain et al. (2018)).

Groups receiving different treatments and the comparison of those means is a post-hoc analysis that often takes place after a general analysis of variance (ANOVA) evaluation. The methods by which means are compared differ depending on factors such as planned versus unplanned comparisons, comparisons between groups of the same or different sizes, comparisons of groups with or without equal variances, and parametric versus stepwise comparisons. However, they have in common the use of statistical distributions such as the F -, t -, or Q -distributions. It should be noted that Day and Quinn (1989), in summary, recommended MCP methods that use the Q -distribution such as the Tukey HSD, REGWQ, and Games-Howell tests. As formulas for this distribution are not available in Excel, it was logical to make a simple user-defined function and implement Studentized range quantiles (Q) in Excel.

K. Thomas Klasson, A Simple but Accurate Excel User-Defined Function to Calculate Studentized Range Quantiles for Tukey Tests

Within this manuscript, the term alpha (α) is used to represent the upper percentile of the Studentized range distribution; α is typically in the range of 0.1 to 0.01 in comparisons. It should be noted that many statistical tables of the Studentized range quantiles (Q) use one minus α (often denoted by p) to identify a table. However, as the upper (rather than lower) percentile (α) is used in several other Excel functions (e.g., FINV and TINV), α was chosen as parameter for the user-defined function. The letter r will be used to denote the total number of groups in the overall experiment and v to denote the degrees of freedom within groups (available in the standard one-way ANOVA table.)

As with many statistical properties related to distributions, statistical tables based on Studentized ranges have long been part of statistical text books but they tend to be limited to a few probabilities (e.g., $p = 0.95$, $\alpha = 0.05$) (Snedecor and Cochran (1967); Walpole and Myers (1989)). For an extensive set of Q tables, see Harter (1960) and Gleason (1998). Lund and Lund (1983) developed a numerical integration algorithm (AS 190) to estimate Q values for a limited range of α -values (0.10-0.01) and also included a rough estimate algebraic algorithm (AS 190.2) for α -values of 0.20-0.05. For a greater range of α , Copenhaver and Holland (1988) developed a Fortran algorithm using numerical integration based on Gauss-Legendre quadrature. The latter algorithm is used by R Statistical Software (Team (2016)) and other statistical software packages. It is, presumably, also the algorithm used by of the Excel add-in RealStats-2007, which is freely available but the algorithm is not viewable (Zaiontz (2007)). While the Copenhaver-Holland iterative method is the most accurate, it requires significant computer processing time and it would be beneficial to have other options in Excel.

Theory and calculations

This manuscript took a non-iterative approach by improving the published AS 190.2 algorithm.

Algorithm AS 190.2 (called QTRNGO) was part of algorithm AS 190 (Lund and Lund (1983)) and is only a

K. Thomas Klasson, A Simple but Accurate Excel User-Defined Function to Calculate Studentized Range Quantiles for Tukey Tests

few lines long. Expressed as an Excel user-defined function, the Excel Visual Basic code for QTRNGO is shown in Figure 1.

```
Function QTRNGO(a, r, v) As Single
'Algorithm AS 190.2, adapted from
'R. E. Lund and J. R. Lund, Algorithm AS 190: Probabilities and upper
'quantiles for the studentized range, Journal of the Royal Statistical Society,
'Series C (Applied Statistics) 32 (1983), no. 2, 204-210.
Dim T, Q As Single
T = Application.WorksheetFunction.NormSInv(0.5 + 0.5 * (1 - a))
If v < 120 Then
T = T + (T ^ 3 + T) / v / 4
End If
Q = 0.8843 - 0.2368 * T
If v < 120 Then
Q = Q - 1.214 / v + 1.208 * T / v
End If
QTRNGO = T * (Q * Log(r - 1) + 1.4142)
End Function
```

Figure 1. Algorithm AS 190.2 (QTRNGO) adapted as a user-defined function in Excel. According to Lund and Lund (1983) this algorithm is suitable for α between 0.01 and 0.2, excluding $v = 2$.

In order to improve the accuracy of AS 190.2, the overall equation for Q in AS 190.2 was expressed as

$$(1) \quad Q = T \cdot [\ln(r-1) \cdot (0.8843 - 0.2368 \cdot T - 1.214 \cdot v^{-1} + 1.208 \cdot T \cdot v^{-1}) + 2^{0.5}] ,$$

where T is a function of α and v with the acknowledgement that T is the equivalent of Excel's $TINV(\alpha, v)$ function. Equation 1 can be rearranged and generalized as

$$(2) \quad W = C_1 \cdot Z_1 + C_2 \cdot Z_2 + C_3 \cdot Z_3 + C_4 \cdot Z_4 ,$$

where

$$W = Q/T - 2^{0.5} , \quad Z_1 = \ln(r-1) , \quad Z_2 = T \cdot \ln(r-1) , \quad Z_3 = v^{-1} \cdot \ln(r-1) , \text{ and } \quad Z_4 = T \cdot v^{-1} \cdot \ln(r-1) .$$

Rather than fitting this equation over a large range of α -values, the equation was fitted to selected values of α which were the same as those published by Gleason (1998). Gleason tabulated Q for $\alpha = 0.5, 0.25, 0.1, 0.05, 0.025, 0.01, 0.005, 0.001$ when $r = 2(1)20(10)40(20)100$ and $v = 2(1)20, 24, 30, 40, 60, 120, \infty$. A typical such table is shown in Figure 2.

K. Thomas Klasson, A Simple but Accurate Excel User-Defined Function to Calculate Studentized Range Quantiles for Tukey Tests

Number of variates (r)														
ν	2	3	4	5	6	7	8	9	10	11	12	13		
1													1	
2	6.085												2	
3	4.501	5.910											3	
4	3.926	5.040	5.757										4	
5	3.635	4.602	5.218	5.673									5	
6	3.460	4.339	4.896	5.305	5.628								6	
7	3.344	4.165	4.681	5.060	5.359	5.606							7	
8	3.261	4.041	4.529	4.886	5.167	5.399	5.596						8	
9	3.199	3.948	4.415	4.755	5.024	5.244	5.432	5.595					9	
10	3.151	3.877	4.327	4.654	4.912	5.124	5.304	5.460	5.598				10	
11	3.113	3.820	4.256	4.574	4.823	5.028	5.202	5.353	5.486	5.605			11	
12	3.081	3.773	4.199	4.508	4.750	4.950	5.119	5.265	5.395	5.510	5.615		12	
13	3.055	3.734	4.151	4.453	4.690	4.884	5.049	5.192	5.318	5.431	5.533	5.625	13	
14	3.033	3.701	4.111	4.407	4.639	4.829	4.990	5.130	5.253	5.364	5.463	5.554	14	
15	3.014	3.673	4.076	4.367	4.595	4.782	4.940	5.077	5.198	5.306	5.403	5.492	15	
16	2.998	3.649	4.046	4.333	4.557	4.741	4.896	5.031	5.150	5.256	5.352	5.439	16	
17	2.984	3.628	4.020	4.303	4.524	4.705	4.858	4.991	5.108	5.212	5.306	5.392	17	
18	2.971	3.609	3.997	4.276	4.494	4.673	4.824	4.955	5.071	5.173	5.266	5.351	18	
19	2.960	3.593	3.977	4.253	4.468	4.645	4.794	4.924	5.037	5.139	5.231	5.314	19	
20	2.950	3.578	3.958	4.232	4.445	4.620	4.768	4.895	5.008	5.108	5.199	5.282	20	
24	2.919	3.532	3.901	4.166	4.373	4.541	4.684	4.807	4.915	5.012	5.099	5.179	24	
30	2.888	3.486	3.845	4.102	4.301	4.464	4.601	4.720	4.824	4.917	5.001	5.077	30	
40	2.858	3.442	3.791	4.039	4.232	4.388	4.521	4.634	4.735	4.824	4.904	4.977	40	
60	2.829	3.399	3.737	3.977	4.163	4.314	4.441	4.550	4.646	4.732	4.808	4.878	60	
120	2.800	3.356	3.685	3.917	4.096	4.241	4.363	4.468	4.560	4.641	4.714	4.781	120	
∞	2.772	3.314	3.633	3.858	4.030	4.170	4.286	4.387	4.474	4.552	4.622	4.685	∞	

Values removed and not used

Number of variates (r)													
ν	14	15	16	17	18	19	20	30	40	60	80	100	ν
1													1
2													2
3													3
4													4
5													5
6													6
7													7
8													8
9													9
10													10
11													11
12													12
13													13
14	5.637												14
15	5.574	5.649											15
16	5.519	5.593	5.662										16
17	5.471	5.544	5.612	5.675									17
18	5.429	5.501	5.567	5.629	5.688								18
19	5.391	5.462	5.528	5.589	5.647	5.701							19
20	5.357	5.427	5.492	5.553	5.610	5.663	5.714						20
24	5.251	5.319	5.381	5.439	5.494	5.545	5.594						24
30	5.147	5.211	5.271	5.327	5.379	5.429	5.475	5.833					30
40	5.044	5.106	5.163	5.216	5.266	5.313	5.358	5.700	5.934				40
60	4.942	5.001	5.056	5.107	5.154	5.199	5.241	5.566	5.789	6.093			60
120	4.842	4.898	4.950	4.998	5.043	5.086	5.126	5.434	5.644	5.929	6.126	6.275	120
∞	4.743	4.796	4.845	4.891	4.934	4.974	5.012	5.301	5.498	5.764	5.947	6.085	∞

Values removed and not used

Figure 2. Q-values as tabulated by Gleason (1998) for $\alpha = 0.05$ with upper right corner removed.

Using the values below the diagonal of the tables, the coefficients of regression C_1 , C_2 , C_3 , and C_4 in Equation 2 can be determined for each value of α using multiple variable least square regression. As shown in Figure 2, only part of the tables were used because it is unlikely that v is less than r in practical situations. For example, consider the comparisons of r groups which has >1 replicates. In the case of 2 replicates, v would take on a value of $r+1$. In fact, as long as all groups contain replicates, v is always greater than r . The results of the regression are shown in Table 1.

Table 1. Coefficients of regression (rounded to 4 decimal points) for Equation 2 for several α -values.

α	C_1	C_2	C_3	C_4	r^2
0.5	392.7272	-580.2495	39.0825	154.6929	0.9998
0.25	157.2291	-136.0289	10.6536	70.9547	0.9998
0.1	40.5373	-24.3509	6.2709	19.6300	0.9998
0.05	16.8278	-8.3896	4.9942	8.3570	0.9998
0.025	7.8603	-3.3642	4.0326	3.9286	0.9998
0.01	3.3294	-1.1908	3.1538	1.6292	0.9998
0.005	1.9070	-0.5972	2.7118	0.8964	0.9998
0.001	0.6344	-0.1380	2.0948	0.2371	0.9997

As is noted in Table 1, the correlation was 0.9997, or better, for each α suggesting very good fit by the relationship in Equation 2. Now, it would be beneficial to evaluate Q at α -values other than those listed in Table 1. It is not recommended to directly interpolate Q in α (Harter (1960)) but Gleason (1999) offered a method of indirect interpolation of Q in α by the correlation

$$(3) \quad Y = m_1 + m_2 \cdot X + m_3 \cdot X^2 ,$$

where

K. Thomas Klasson, A Simple but Accurate Excel User-Defined Function to Calculate Studentized Range Quantiles for Tukey Tests

$$(4, 5) \quad Y = \ln(Q_{\alpha,r,v} \cdot 2^{-0.5} \cdot T_{\alpha,v}^{-1} + r \cdot v^{-1}) \quad \text{and} \quad X = -(1 + 1.5 \cdot N_{1-\alpha/2})^{-1}.$$

The entities $T_{\alpha,v}$ and $N_{1-\alpha/2}$ are equal to Excel's functions $\text{TINV}(\alpha,v)$ and $\text{NORMSINV}(1-\alpha/2)$. Thus, with a known r - and v -value, we can calculate Y - and X -values for each of the α -values in Table 1. This will allow us to determine m_1 , m_2 , and m_3 in Equation 3. With these constants and the X -value for the desired α (via Equation 5), we can calculate the Y -value through Equation 3 which allow us to back-calculate the Q -value via Equation 4. This method of interpolation can also be used for extrapolation for α -values below those listed in Table 1 (Gleason (1999)). When the method for calculating Q was initially tested, the results indicated that when $v > 100$ and $\alpha > 0.025$ the method under-estimated the Q -value with 0.7%. This small adjustment was added to the computer code. In addition, the error of the calculated Q -values followed a complicated pattern as a function of r but it appeared that the errors for $r \geq 20$ could be addressed, as the error was linear with r . This was found true for all α -values and the slopes and intercepts were linearly dependent on $\ln(\alpha)$. Thus, this small correction term was also added to the computer code to adjust Q -values when $r \geq 20$. The final Excel Visual Basic code for the user defined function $\text{QTAB}(\alpha,r,v)$ is shown in Figure 3.

K. Thomas Klasson, A Simple but Accurate Excel User-Defined Function to Calculate Studentized Range Quantiles for Tukey Tests

```

Public Function QTAB(a, r, v) As Single
Dim i, j, k, Done As Integer
Dim alpha(8), m(3, 4), X(8), Y(8), C(4, 8), Q As Single

'If outside range of use, report #VALUE error and quit
If r < 2 Or r > 100 Or v < r Or a < 0.0001 Or a > 0.5 Then
    QTAB = 1 / 0
    GoTo Done
End If

alpha(1) = 0.5: C(1, 1) = 392.727161: C(2, 1) = -580.249479: C(3, 1) = 39.08247679: C(4, 1) = 154.6928606
alpha(2) = 0.25: C(1, 2) = 157.2290812: C(2, 2) = -136.0288671: C(3, 2) = 10.65361764: C(4, 2) = 70.95473271
alpha(3) = 0.1: C(1, 3) = 40.53727: C(2, 3) = -24.35088143: C(3, 3) = 6.270918133: C(4, 3) = 19.63004347
alpha(4) = 0.05: C(1, 4) = 16.82778272: C(2, 4) = -8.38960093: C(3, 4) = 4.994168622: C(4, 4) = 8.357020633
alpha(5) = 0.025: C(1, 5) = 7.860314249: C(2, 5) = -3.564167121: C(3, 5) = 4.032604321: C(4, 5) = 3.928646444
alpha(6) = 0.01: C(1, 6) = 3.329358: C(2, 6) = -1.190778187: C(3, 6) = 3.153807539: C(4, 6) = 1.629208317
alpha(7) = 0.005: C(1, 7) = 1.907028105: C(2, 7) = -0.597201494: C(3, 7) = 2.7117957: C(4, 7) = 0.896448643
alpha(8) = 0.001: C(1, 8) = 0.634353634: C(2, 8) = -0.137994025: C(3, 8) = 2.094748: C(4, 8) = 0.237087945

'Check to see if a is equal to one of the standard alpha values
j = 0: Done = 0
For i = 1 To 8
    If a = alpha(i) Then
        j = i: Done = 1
    End If
Next i

'If we do not need interpolation/extrapolation in alpha, calculate the Q-value for a=alpha(j) and exit
If Done = 1 Then
    QTAB = Application.WorksheetFunction.TInv(a, v) * (Sqr(2) + Log(r - 1) * (C(1, j) + C(3, j) / v) + _
        Application.WorksheetFunction.TInv(a, v) * Log(r - 1) * (C(2, j) + C(4, j) / v))
    GoTo Done:
End If

'INTERPOLATION AND EXTRAPOLATION IN alpha
'For each alpha value, calculate Y- and X-values which will be used to interpolate/extrapolate
For i = 1 To 8
    Q = Application.WorksheetFunction.TInv(alpha(i), v) * (Sqr(2) + Log(r - 1) * (C(1, i) + C(3, i) / v) + _
        Application.WorksheetFunction.TInv(alpha(i), v) * Log(r - 1) * (C(2, i) + C(4, i) / v))
    Y(i) = Log(Q / Sqr(2)) / Application.WorksheetFunction.TInv(alpha(i), v) + r / v
    X(i) = -1 / (1 + 1.5 * Application.WorksheetFunction.NormSInv(1 - alpha(i) / 2))
Next i

'Form the matrix that is needed to find coefficients of regression in quadratic equation with Y=fn(X)
For i = 1 To 8
    m(1, 1) = m(1, 1) + 1: m(1, 2) = m(1, 2) + X(i): m(1, 3) = m(1, 3) + X(i) ^ 2: m(1, 4) = m(1, 4) + Y(i)
    m(2, 3) = m(2, 3) + X(i) ^ 3: m(2, 4) = m(2, 4) + X(i) * Y(i)
    m(3, 3) = m(3, 3) + X(i) ^ 4: m(3, 4) = m(3, 4) + X(i) ^ 2 * Y(i)
Next i
m(2, 1) = m(1, 2): m(2, 2) = m(1, 3): m(3, 1) = m(1, 3): m(3, 2) = m(2, 3)

'Solve matrix using Gaussian elimination
For i = 1 To 2
    For j = i + 1 To 3
        For k = j + 1 To 4
            m(j, k) = m(j, k) - m(i, k) * m(j, i) / m(i, i)
        Next k
    Next j
Next i
m(3, 4) = m(3, 4) / m(3, 3)
m(2, 4) = (m(2, 4) - m(2, 3) * m(3, 4)) / m(2, 2)
m(1, 4) = (m(1, 4) - m(1, 2) * m(2, 4) - m(1, 3) * m(3, 4)) / m(1, 1)
'm(1,4), m(2,4), and m(3,4) are the coefficients of regression in Y=m(1,4)+m(2,4)*X+m(3,4)*X^2

'Calculate X-value for the desired alpha and use the polynomial to find its Y-value.
X(0) = -1 / (1 + 1.5 * Application.WorksheetFunction.NormSInv(1 - a / 2))
Y(0) = m(1, 4) + m(2, 4) * X(0) + m(3, 4) * X(0) ^ 2

'Calculate the interpolated/extrapolated Q-value
QTAB = (Exp(Y(0)) - r / v) * Sqr(2) * Application.WorksheetFunction.TInv(a, v)

Done:
'Adjust Q-value if v>100, a>0.025 and if r>=20
If v > 100 And a > 0.025 Then
    QTAB = QTAB / (1 - 0.007)
End If
If r > 19 Then
    QTAB = QTAB / (1 + ((0.0000769 * Log(a) + 0.000605) * r + (-0.000625 * Log(a) - 0.00976)))
End If
End Function

```

Figure 3. The user-defined function $QTAB(\alpha, r, v)$ implemented in a Visual Basic module of Excel. Note

that the natural logarithm function in Excel Visual Basic is Log.

Results

The accuracy of the QTRNGO (Figure 1) and QTAB (Figure 3) functions is shown in Table 2. As is noted, the QTAB function was considerably more accurate with less than 0.7% absolute mean error and could be used over a very large range of α -values. The difference in fit between QTRNGO and QTAB is also shown in Figure 4 for $\alpha = 0.05$. Again, the improvement in accuracy by QTAB over QTRNGO is clear. As far as computational efficiency, the 960 α -values used for error calculations of the interpolation and extrapolation results in Table 2 took 489 s to calculate with the RealStats-2007's add-in, which presumably uses the Copenhaver-Holland technique (Zaiontz (2007)). The same 960 calculations with QTRNGO and QTAB took 1.5 and 2.5 s, respectively. If a higher accuracy is desired, a more complex user-defined function has also been published by Klasson (2018), which took a different approach than the one developed here

It is anticipated that this user-defined function will help researchers and student who want to use Excel to conduct fast multiple comparisons procedures after initial ANOVA tests. The accuracy, computational efficiency, large usable range of variables, and simple programing of the method are beneficial. The QTAB function forces a #VALUE error if $r < 2$, $r > 100$, $v < r$, $\alpha < 0.0001$, or $\alpha > 0.5$.

Table 2. Accuracy of two Excel user-defined functions for $Q_{\alpha,r,v}$. The accurate values for the top half of the table was from Gleason (1998) and the accurate values for the bottom half of the table was generated by the QINV function of RealStat-2007 (Zaiontz (2007)).

α	QTRNGO			QTAB		
	Max.	Min.	Abs. Mean	Max.	Min.	Abs. Mean
	Error	Error	Error	Error	Error	Error
0.5	0.0%	-36.6%	29.6% ^a	1.4%	-1.4%	0.42%
0.25	0.0%	-14.8%	29.6% ^a	0.9%	-1.3%	0.32%
0.1	4.4%	-8.5%	2.0%	1.0%	-0.9%	0.25%
0.05	7.7%	-13.6%	1.0%	1.1%	-0.9%	0.27%
0.025	7.5%	-19.4%	1.2%	1.1%	-0.9%	0.33%
0.01	3.8%	-27.8%	1.5%	1.2%	-0.9%	0.33%
0.005	0.0%	-34.5%	3.7% ^a	1.2%	-0.9%	0.34%
0.001	0.0%	-49.5%	3.7% ^a	1.3%	-0.8%	0.37%
Interpolation and Extrapolation						
0.15	0.0%	-7.2%	4.8%	0.8%	-1.1%	0.30%
0.075	6.3%	-11%	1.0%	1.0%	-0.8%	0.26%
0.0001	0.0%	-66%	25% ^a	5.4% ^b	-2.2%	0.69%

^aQTRNGO (AS 190.2) is not recommended for these α -values (Lund and Lund (1983)). ^bThe Excel add-in RealStats-2007 reported an unreasonable low value for $Q_{0.0001,2,2}$ and this data point was removed from the error calculations. The web site for the RealStats-2007 add-in does not recommend calculating Q for low α -values (Zaiontz (2007)) which may be the cause for the slightly greater errors for $\alpha = 0.0001$.




Figure 4. The accuracy of the two fast estimation methods for Q when $\alpha = 0.05$. The values from the statistical table were taken from Gleason (1998).

Acknowledgements

Mention of trade names or commercial products in this publication is solely for the purpose of providing specific information and does not imply recommendation or endorsement by the U.S. Department of Agriculture. USDA is an equal opportunity provider and employer.

References

- Copenhaver, MDHolland, B: Computation of the distribution of the maximum studentized range statistic with application to multiple significance testing of simple effects. *Journal of Statistical Computation and Simulation* **30**(1), 1-15 (1988).
- Day, RWQuinn, GP: Comparisons of treatments after an analysis of variance in ecology. *Ecological Monographs* **59**(4), 433-463 (1989).
- Franklin, HM, Garzon-Garcia, A, Burton, J, Moody, PW, De Hayr, RWBurford, MA: A novel bioassay to assess phytoplankton responses to soil-derived particulate nutrients. *Science of the Total Environment* **636**, 1470-1479 (2018).

K. Thomas Klasson, A Simple but Accurate Excel User-Defined Function to Calculate Studentized Range Quantiles for Tukey Tests

Gleason, JR: dm64: Quantiles of the Studentized range Distribution. Stata Technical Bulletin **46**, 6-10 (1998).

Gleason, JR: An accurate, non-iterative approximation for studentized range quantiles. Computational Statistics and Data Analysis **31**(2), 147-158 (1999).

Harter, HL: Tables of Range and Studentized Range. The Annals of Mathematical Statistics **31**(4), 1122-1147 (1960).

Klasson, KT: QXLA: Adding upper quantiles for the Studentized range to Excel for multiple comparison procedures. Journal of Statistical Software **85**(Code Snippet 1), 1-9 (2018).

Kramer, MH, Paparozzi, ET, Stroup, WW: Statistics in a horticultural journal: Problems and solutions. Journal of the American Society for Horticultural Science **141**(5), 400-406 (2016).

Lund, RE, Lund, JR: Algorithm AS 190: Probabilities and Upper Quantiles for the Studentized Range. Journal of the Royal Statistical Society Series C (Applied Statistics) **32**(2), 204-210 (1983).

Schroeder-Spain, K, Fisher, LLS, Mee, DL: Uncoordinated: Effects of sublethal malathion and carbaryl exposures on juvenile and adult blue crabs (*Callinectes sapidus*). Journal of Experimental Marine Biology and Ecology **504**, 1-9 (2018).

Snedecor, GW, Cochran, WG: Statistical Methods. The Iowa State University Press, Ames, IA (1967).

Team, RC (2016). R: A Language and Environment for Statistical Computing. Vienna, Australia, R Foundation for Statistical Computing.

Walpole, RE, Myers, RH: Probability and Statistics for Engineers and Scientists. Macmillan Publishing, New York, NY (1989).

Zaiontz, C: Real statistics using Excel. <http://real-statistics.com> (2007). Accessed June 24.